

Interpolazione Polinomiale soluzione diretta delle condizioni di interpolazione

Enrico Bertolazzi

```
> # carica la libreria per l'algebra lineare  
with(LinearAlgebra):  
> # definisco i dati del problema  
# N = numero di punti  
# X = coordinate x dei punti da interpolare  
# Y = coordinate y dei punti da interpolare  
N,X,Y := 4, <1,3,5/2,4>, <2,2,4,4> ;
```

$$N, X, Y := 4, \begin{bmatrix} 1 \\ 3 \\ \frac{5}{2} \\ 4 \end{bmatrix}, \begin{bmatrix} 2 \\ 2 \\ 4 \\ 4 \end{bmatrix} \quad (1)$$

```
> # VR costruisce la riga della matrice di Vandermonte  
VR := x -> <1,x,x^2,x^3> ;
```

$$VR := x \rightarrow rtable/ConstructColumn(1, x, x^2, x^3) \quad (2)$$

```
> # posso costruire VR in modo che dipenda da N  
VR := x -> <seq(x^i,i=0..N-1)> ;
```

$$VR := x \rightarrow rtable/ConstructColumn(seq(x^i, i = 0 .. N - 1)) \quad (3)$$

```
> # alcuni esempi  
VR(z), VR(2), VR(w*x/5) ;
```

$$\begin{bmatrix} 1 \\ z \\ z^2 \\ z^3 \end{bmatrix}, \begin{bmatrix} 1 \\ 2 \\ 4 \\ 8 \end{bmatrix}, \begin{bmatrix} 1 \\ \frac{1}{5} w x \\ \frac{1}{25} w^2 x^2 \\ \frac{1}{125} w^3 x^3 \end{bmatrix} \quad (4)$$

```
> # costruisco la matrice di Vandermonte  
# per l'interpolazione  
M := Transpose(<VR(X[1])|VR(X[2])|VR(X[3])|VR(X[4])>);
```

(5)

$$M := \begin{bmatrix} 1 & 1 & 1 & 1 \\ 1 & 3 & 9 & 27 \\ 1 & \frac{5}{2} & \frac{25}{4} & \frac{125}{8} \\ 1 & 4 & 16 & 64 \end{bmatrix} \quad (5)$$

> # posso costruire la matrice M in modo dipendente da N
M := <seq(Transpose(VR(X[i])),i=1..N)>;

$$M := \begin{bmatrix} 1 & 1 & 1 & 1 \\ 1 & 3 & 9 & 27 \\ 1 & \frac{5}{2} & \frac{25}{4} & \frac{125}{8} \\ 1 & 4 & 16 & 64 \end{bmatrix} \quad (6)$$

> # definisco il termine noto del problema
b := <Y[1],Y[2],Y[3],Y[4]> ;

$$b := \begin{bmatrix} 2 \\ 2 \\ 4 \\ 4 \end{bmatrix} \quad (7)$$

> # posso costruire il vettore b in modo dipendente da N
b := <seq(Y[i],i=1..N)> ;

$$b := \begin{bmatrix} 2 \\ 2 \\ 4 \\ 4 \end{bmatrix} \quad (8)$$

> # risolvo il sistema l;ineare invertendo la matrice M
a := M^(-1).b ;

$$a := \begin{bmatrix} -\frac{68}{3} \\ \frac{356}{9} \\ -\frac{154}{9} \\ \frac{20}{9} \end{bmatrix} \quad (9)$$

> # in alternativa posso usare il solutore
lineare di Maple

```
a := LinearSolve(M,b) ;
```

$$a := \begin{bmatrix} -\frac{68}{3} \\ \frac{356}{9} \\ -\frac{154}{9} \\ \frac{20}{9} \end{bmatrix} \quad (10)$$

```
> # costruisco il polinomio interpolante
```

```
p := a[1] + a[2]*x + a[3] * x^2 + a[4] * x^3 ;
```

$$p := -\frac{68}{3} + \frac{356}{9}x - \frac{154}{9}x^2 + \frac{20}{9}x^3 \quad (11)$$

```
> # in alternativa lo costruisco in modo dipendente  
# da N col comando add
```

```
p := add(a[i]*x^(i-1),i=1..N) ;
```

$$p := -\frac{68}{3} + \frac{356}{9}x - \frac{154}{9}x^2 + \frac{20}{9}x^3 \quad (12)$$

```
> # controllo che il polinomio calcolato interpoli  
# effettivamente i punti dati (sempre verificare!)
```

```
seq(subs(x=X[i],p),i=1..N);  
seq(Y[i],i=1..N);
```

2, 2, 4, 4


2, 2, 4, 4

(13)


```
> # richiamo la libreria per il plottaggio  
with(plots):
```

```
Warning, the name changecoords has been redefined
```


```
> # plottaggio del polinomio interpolante  
plot(p,x=0..5);
```



```
> # plottaggio dei punti di interpolazione  
plot([X[1],Y[1]],[X[2],Y[2]],[X[3],Y[3]],[X[4],Y[4]]],  
      style=POINT, # plottaggio di punti non linee  
      symbolsize=50, # dimensione del simbolo  
      color=blue);
```



```
> # stessa cosa ma dipendente da N
plot([seq([X[i],Y[i]],i=1..N)],
     style=POINT,
     symbolsize=50,
     color=blue);
```



```
> # per combinare i plottaggi li salvo sulle
# variabili A e B
A := plot(p,x=0..5):
B := plot([seq([X[i],Y[i]],i=1..N)],
         style=POINT,
         symbolsize=50,
         color=blue):
> # e con il comando display li combino
display({A,B});
```

